Adaptive dynamics models

Logistic growth model

Unconstrained evolution of birth rate

What we are asking here is whether a mutant host with a different birth rate could invade when the resident host is at its carrying capacity. Mathematically, this is equivalent to asking whether the Jacbian matrix, evaluated at the mutant-free equilibrium, has at least one positive eigenvalue.

In[8]:=
```
(* This is the model *)
dR = b (1 - (R + Rm)/K) R - d (1 - (R + Rm)/K) R;
dRm = bm (1 - (R + Rm)/K) Rm - d (1 - (R + Rm)/K) Rm;
(* The Jacobian *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}]]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} -\frac{b\,R}{K}+\frac{d\,R}{K}+b\left(1-\frac{R+Rm}{K}\right)-d\left(1-\frac{R+Rm}{K}\right) & -\frac{b\,R}{K}+\frac{d\,R}{K} \\ -\frac{bm\,Rm}{K}+\frac{d\,Rm}{K} & -\frac{bm\,Rm}{K}+\frac{d\,Rm}{K}+bm\left(1-\frac{R+Rm}{K}\right)-d\left(1-\frac{R+Rm}{K}\right) \end{pmatrix}$$

In[11]:=
```
(* The Jacobian, evaluated at the mutant-free equilibrium *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}] /. {R → K, Rm → 0}]
```

Out[11]//MatrixForm=

$$\begin{pmatrix} -b+d & -b+d \\ 0 & 0 \end{pmatrix}$$

In[12]:=
```
(* Eigenvalues are *)
Eigenvalues[J]
```

Out[12]= $\{-b+d, 0\}$

Based on this, it is clear that -b+d < 0, which is necessary for the resident to be able to exist at all. The 0 eigenvalue implies that the system is neutrally stable, and no evolution will happen in this system.

Unconstrained evolution of carrying capacity

What we are asking here is whether a mutant host with a different carrying capacity could invade when the resident host is at its carrying capacity.

In[56]:=
```
(* This is the model *)
dR = b (1 - (R + Rm)/K) R - d (1 - (R + Rm)/K) R;
dRm = b (1 - (R + Rm)/Km) Rm - d (1 - (R + Rm)/Km) Rm;
(* The Jacobian *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}]]
```

Out[58]//MatrixForm=

$$\begin{pmatrix} -\frac{b R}{K} + \frac{d R}{K} + b \left(1 - \frac{R+Rm}{K}\right) - d \left(1 - \frac{R+Rm}{K}\right) & -\frac{b R}{K} + \frac{d R}{K} \\ -\frac{b\, Rm}{Km} + \frac{d\, Rm}{Km} & -\frac{b\, Rm}{Km} + \frac{d\, Rm}{Km} + b \left(1 - \frac{R+Rm}{Km}\right) - d \left(1 - \frac{R+Rm}{Km}\right) \end{pmatrix}$$

In[59]:=
```
(* The Jacobian, evaluated at the mutant-free equilibrium *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}] /. {R → K, Rm → 0}]
```

Out[59]//MatrixForm=

$$\begin{pmatrix} -b + d & -b + d \\ 0 & b \left(1 - \frac{K}{Km}\right) - d \left(1 - \frac{K}{Km}\right) \end{pmatrix}$$

In[60]:=
```
(* Eigenvalues are *)
Eigenvalues[J] // Simplify
```

Out[60]= $\left\{-b + d, \frac{(b - d)\ (-K + Km)}{Km}\right\}$

The second eigenvalue is termed the *invasion fitness* for this model. If it is positive, then the mutant can invade (and is assumed to displace the resident). In this case, the invasion fitness will be positive whenever $K_m > K$, exactly as one might expect.

To determine how evolution will play out, we compute the *fitness gradient*, which is the derivative of the invasion fitness with respect to the evolving trait. The fitness gradient for this model is the following (note that convention in adaptive dynamics is to evaluate the fitness gradient at the point where the mutant and resident are identical in their traits, reflective the adaptive dynamics assumption of small mutations):

In[61]:=
```
D[Eigenvalues[J][[2]], Km] /. {Km → K} // FullSimplify
```

Out[61]= $\frac{b - d}{K}$

The fitness gradient is always positive in this unconstrained model, implying that evolution will drive carrying capacity to infinity.

Constrained evolution of carrying capacity

In[84]:=
```
(* This is the model *)
dR = b[K] (1 - (R + Rm)/K) R - d (1 - (R + Rm)/K) R;
dRm = b[Km] (1 - (R + Rm)/Km) Rm - d (1 - (R + Rm)/Km) Rm;
(* The Jacobian *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}]]
```

In[87]:=
```
(* The Jacobian, evaluated at the mutant-free equilibrium *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}] /. {R → K, Rm → 0}]
```

In[88]:=
```
(* Eigenvalues are *)
Eigenvalues[J] // Simplify
```

The fitness gradient for this model is the following (note that convention in adaptive dynamics is to evaluate the fitness gradient at the point where the mutant and resident are identical in their traits, reflective the adaptive dynamics assumption of small mutations):

In[89]:=
```
D[Eigenvalues[J][[2]], Km] /. {Km → K} // FullSimplify
```

Out[89]=

$$\frac{-d + b[K]}{K}$$

Thus, there will be an evolutionarily stable strategy for this model, which will occur exactly when $b(K) = d$.Thus, from any starting carrying capacity, evolution will increase carrying capacity until birth and death rates are balanced.

Alternative logistic growth model

Unconstrained evolution of birth rate

In[62]:=
```
(* This is the model *)
dR = (b - bs (R + Rm)) R - (d + ds (R + Rm)) R;
dRm = (bm - bs (R + Rm)) Rm - (d + ds (R + Rm)) Rm;
(* The Jacobian *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}]] // Simplify
```

Out[64]//MatrixForm=

$$\begin{pmatrix} b - d - (bs + ds)(2R + Rm) & -(bs + ds) R \\ -(bs + ds) Rm & bm - d - (bs + ds)(R + 2Rm) \end{pmatrix}$$

In[65]:=
```
(* The Jacobian, evaluated at the mutant-free equilibrium *)
Solve[(dR /. Rm → 0) == 0, R]
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}] /. {R → (b - d)/(bs + ds), Rm → 0}]
```

Out[65]= $\left\{\{R \to 0\}, \left\{R \to \frac{b-d}{bs+ds}\right\}\right\}$

Out[66]//MatrixForm=

$$\begin{pmatrix} b - d - \frac{2\,bs\,(b-d)}{bs+ds} - \frac{2\,(b-d)\,ds}{bs+ds} & -\frac{bs\,(b-d)}{bs+ds} - \frac{(b-d)\,ds}{bs+ds} \\ 0 & bm - d - \frac{bs\,(b-d)}{bs+ds} - \frac{(b-d)\,ds}{bs+ds} \end{pmatrix}$$

In[67]:=
```
(* Eigenvalues are *)
Eigenvalues[J] // Simplify
```

Out[67]= $\{-b + bm, -b + d\}$

The invasion fitness is $b_m - b$, implying that the mutant can invade if it has a higher birth rate than the resident. This difference with the first model is because here, birth rate affects the carrying capacity as well.

The fitness gradient is always positive, implying, as expected, that evolution will drive *b* to infinity.

In[69]:=
```
D[Eigenvalues[J][[1]], bm]
```

Out[69]= 1

Constrained evolution of birth rate

Here I assume that d is a function of b (keeping it unspecified for the moment as well).

In[70]:=
```
(* This is the model *)
dR = (b - bs (R + Rm)) R - (d[b] + ds (R + Rm)) R;
dRm = (bm - bs (R + Rm)) Rm - (d[bm] + ds (R + Rm)) Rm;
(* The Jacobian *)
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}]] // Simplify
```

Out[72]//MatrixForm=

$$\begin{pmatrix} b - (bs + ds)\,(2\,R + Rm) - d[b] & -(bs + ds)\,R \\ -(bs + ds)\,Rm & bm - (bs + ds)\,(R + 2\,Rm) - d[bm] \end{pmatrix}$$

In[75]:=
```
(* The Jacobian, evaluated at the mutant-free equilibrium *)
Solve[(dR /. Rm → 0) == 0, R]
MatrixForm[J = D[{dR, dRm}, {{R, Rm}}] /. {R → (b - d[b])/(bs + ds), Rm → 0}]
```

Out[75]= $\left\{\{R \to 0\}, \left\{R \to \frac{b-d[b]}{bs+ds}\right\}\right\}$

Out[76]//MatrixForm=

$$\begin{pmatrix} b - \frac{2\,bs\,(b-d[b])}{bs+ds} - \frac{2\,ds\,(b-d[b])}{bs+ds} - d[b] & -\frac{bs\,(b-d[b])}{bs+ds} - \frac{ds\,(b-d[b])}{bs+ds} \\ 0 & bm - \frac{bs\,(b-d[b])}{bs+ds} - \frac{ds\,(b-d[b])}{bs+ds} - d[bm] \end{pmatrix}$$

In[77]:=
```
(* Eigenvalues are *)
Eigenvalues[J] // Simplify
```

Out[77]= {-b + d[b], -b + bm + d[b] - d[bm]}

The invasion fitness is $b_m - b + d(b) - d(b_m)$. Notice that if $b_m > b$, then $d(b_m) > d(b)$ suggesting that the invasion fitness may not always be positive. As such, there are more possibilities for evolution in this system.

The fitness gradient bear this out. It is $1 - d'(b)$; since $d'(b) > 0$, the fitness gradient may be positive or negative, depending on whether, at the current b, $d'(b) < 1$ or not. In this case, there will be a *singular strategy* (an endpoint of evolution) when $d'(b) = 1$. In other words, evolution will continue until b satisfies this condition. Notice that a linear function will never be able to satisfy this condition - $d(b)$ is linear then $d'(b)$ will be a constant that will either be larger or smaller than 1, and evolution will either drive b to zero or to infinity, respectively.

In[79]:=
```
D[Eigenvalues[J][[2]], bm] /. bm → b
```

Out[79]= 1 - d′[b]

However, if $d(b)$ is nonlinear, then singular strategies are possible. Whether any singular strategy represents an *evolutionarily stable strategy* (a fitness maximum) depends on the second derivative of the fitness gradient with respect to the evolving trait, evaluated at the singular strategy. (I.e., it is the curvature of the fitness gradient; if the second derivative is negative, the fitness gradient is at a peak.) This ESS condition is $-d''(b)$, meaning that $d''(b) > 0$ for there to be an evolutionarily stable strategy (i.e., death rate must be accelerating function of birth rate).

In[80]:=
```
D[Eigenvalues[J][[2]], {bm, 2}] /. bm → b
```

Out[80]= -d″[b]

Here’s a numerical example, just to make this a bit more concrete, and to potentially give you something to push back against with the GEM.

If $d(b) = d_0 b^2$, with $d_0 = 0.1$, then the evolutionarily stable b value will be:

In[96]:=
```
Solve[2 d0 b == 1, b]
Solve[2 d0 b == 1, b] /. d0 → 0.1
```

Out[96]= $\left\{\left\{b \to \frac{1}{2\ d0}\right\}\right\}$

Out[97]= {{b → 5.}}

The carrying capacity at this evolutionarily stable b value (assuming $b_{slope} = d_{slope} = 0.1$ is 12.5.

In[103]:=
```
Solve[(b - bs (R)) R - (d[b] + ds (R)) R == 0, R][[2]] /. {d[b] → d0 b^2} /. {b → 1/(2 d0)} /.
  {d0 → 0.1, bs → 0.1, ds → 0.1}
```

Out[103]= {R → 12.5}

Epidemiological model

We'll go with a fairly simple model here, just to illustrate the possibilities. Since I already know what happens in the absence of any trade-offs, I will assume that the transmission rate, β, is an increasing function of virulence, and we will try to find the evolutionarily stable virulence strategy. I will assume that there is a novel parasite with a different virulence that infects susceptible hosts, giving rise to hosts infected with the mutant parasite (I_m), compared to the hosts infected with the resident parasite (I). I assume that all infected hosts give birth to susceptible hosts. For utter simplicity, I'll just assume exponential growth in the absence of the parasite.

In[129]:=
```
(* the model *)
dS = r (S + I + Im) - β[v] S I - β[vm] S Im - m S;
dI = β[v] S I - (m + v) I;
dIm = β[vm] S Im - (m + vm) Im;
(* the Jacobian *)
MatrixForm[J = D[{dS, dI, dIm}, {{S, I, Im}}] /. {Im → 0, S → (m + v)/β[v]}]
```

Out[132]//MatrixForm=

$$\begin{pmatrix} -m + r - I\,\beta[v] & -m + r - v & r - \frac{(m+v)\,\beta[vm]}{\beta[v]} \\ I\,\beta[v] & 0 & 0 \\ 0 & 0 & -m - vm + \frac{(m+v)\,\beta[vm]}{\beta[v]} \end{pmatrix}$$

The mutant can invade if $(m+v)\,\frac{\beta(v_m)}{\beta(v)} - (m - v_m) > 0$. The fitness gradient is

In[134]:=
```
D[-m - vm + ((m + v) β[vm])/β[v], vm] /. vm → v
```

Out[134]= $-1 + \frac{(m + v)\,\beta'[v]}{\beta[v]}$

Given this fitness gradient, there will be a singular strategy. It will be evolutionarily stable if $\beta''(v) < 0$, that is, if transmission is a saturating function of virulence.

In[135]:=
```
D[-m - vm + ((m + v) β[vm])/β[v], {vm, 2}] /. vm → v
```

Out[135]= $\frac{(m + v)\,\beta''[v]}{\beta[v]}$

So, if $\beta(v) = \beta_0\,\frac{v}{1+v}$, then the evolutionarily stable strategy will be $v = \sqrt{m}$.

In[137]:=
```
Solve[(-1 + ((m + v) β'[v])/β[v] /. {β[v] → β0 v/(1 + v)} /. {β'[v] → D[β0 v/(1 + v), v]}) == 0, v]
```

Out[137]= $\{\{v \to -\sqrt{m}\}, \{v \to \sqrt{m}\}\}$

So if m = 1, then the evolutionarily stable v will also be 1. The equilibrium abundance of susceptible and infected hosts at the evolutionarily stable virulence is given below, given $\beta_0 = 0.1$, $r = 1.5$, is $\hat{S} = 40$ and $\hat{I} = 40$.

In[165]:=
```
dS = r (S + I) - β0 v/(1 + v) S I - m S;
dI = β0 v/(1 + v) S I - (m + v) I;
Solve[{dS == 0, dI == 0}, {S, I}]
Solve[{dS == 0, dI == 0}, {S, I}][[2]] /. {m → 1, v → 1, r → 1.5, β0 → 0.1}
```

Out[167]= $\left\{\{S \to 0, I \to 0\}, \left\{S \to \frac{(1+v)\ (m+v)}{v\ \beta 0}, I \to -\frac{(m-r)\ (1+v)\ (m+v)}{v\ (m-r+v)\ \beta 0}\right\}\right\}$

Out[168]= $\{S \to 40., I \to 40.\}$